Atomic & molecular cross-section and collision rates data

Vladislav Kotov, FZJ

November, 2016

Inhaltsverzeichnis

1 Fit formulas

Interpolation polynomials are introduced in the book [Janev, R.K., Langer, W.D., Evans, K.J., Post, D.E.J., Elementary Processes in Hydrogen-Helium Plasmas, 1987]. See also `http://www.eirene.de/janev_buch.pdf`, Section 9. There are two types of the fits - single polynomial and double polynomial:

$$\ln Y = \sum_{i=0}^{N} A_i \left(\ln X\right)^i \tag{1}$$

$$\ln Y = \sum_{i=0}^{N} \sum_{j=0}^{M} A_{ji} \left(\ln X_1\right)^i \left(\ln X_2\right)^j \tag{2}$$

First equation is used to interpolate function $Y\left(X\right)$, second equation is applied for functions of two variables $Y\left(X_1, X_2\right)$. Coefficients of the fit formulas for individual reactions are stored in the data files: AMJUEL, HYDHEL, METHANE, H2vibr. Most of the fits in those files have $N = 8$, $M = 8$. The data sets were originally created and collected for the particle transport code EIRENE, see `www.eirene.de`.

2 Types of reactions

Type	Result	1st variable	2nd variable
H.1	Cross Section [cm^2]	Energy [eV]	-
H.2	Collision Rate [cm^3/s]	Temperature [eV]	-
H.3	Collision Rate [cm^3/s]	Energy [eV]	Temperature [eV]
H.4	Collision Rate [cm^3/s]	Density* [cm^{-3}]	Temperature [eV]
H.8	Electron cooling rate [$cm^3/s{\cdot}eV$]	Temperature [eV]	-
H.9	Ion cooling rate [$cm^3/s{\cdot}eV$]	Energy [eV]	Temperature [eV]
H.10	Electron cooling rate [$cm^3/s{\cdot}eV$]	Density* [cm^{-3}]	Temperature [eV]
H.11	Ratio	Temperature [eV]	-
H.12	Ratio	Density* [cm^{-3}]	Temperature [eV]

* density inserted into the fit formulas must be the density in cm^{-3} **divided by 1e8**. In the interface `eiram_matlab_calc2` this transformation is made automatically for reaction types H.4,10,12. Therefore, one does not have to divide by 1e8 when using the Matlab/Octave and CGI (web) interfaces.

3 The validity range issue

Polynomial fits can only be applied in the parameter range for which they were generated. This constraint has to be taken very strictly because fits applied for parameters (X_1, X_2) lying outside of range can produce completely meaningless results even if deviation from the bounds is small. Unfortunately, for many reactions in the EIRENE data files the information about the boundaries of the validity domain is missing. Efforts are ongoing to restore this information and to make it available.

Most of the fits are made for the energy and temperature range **from 0.1 eV to 1e4 eV**. One can not assume, however, that this validity range applies to all reactions. Heavy particle collisions in AMJUEL, for example, seem to be defined for energies and temperatures $\leq$1e3 eV. The density range is typically **from 1e8 cm^{-3} to 1e16 cm^{-3}**.

4 Example of usage

4.1 Makefile

```
SHELL = /bin/sh
FC=gfortran -Wall -Wextra -fcheck=all

run : eiram_example.exe
	./eiram_example.exe

eiram_example.exe: eiram_example.f
	@echo EIRAM must be installed in '../bin'
	$(FC) -o $@ eiram_example.f -L../bin -leiram -I../bin

clean:
	rm *.o *.exe
```

4.2 Program

```
C> @file eiram_example.f90
C> Example of usage for EIRAM
C> @author Vladislav Kotov, v.kotov@fz-juelich.de
      program eiram_example
      use eiram

      integer :: err,id,k,N,M
      integer,parameter :: L=4
      real(eiram_dp) :: Eb(L)=(/1,10,100,1000/),
     r                  Ti(L)=(/1,10,100,1000/),
     r                  ne(L)=(/1e15,1e14,1e13,1e12/),
     r                  sigma(L),rate(L),lnE
      real(eiram_dp),allocatable :: B(:,:)

      intrinsic :: log,exp

C ######################################################################
C  Calculation of cross-section (depend only on energy)
C ######################################################################

C     load HYDHEL
      call eiram_load('../data/','HYDHEL',err)
      if(err.ne.0) STOP "ERROR while loading HYDHEL"

C     take Id of reaction H.1 3.1.8 -
C     - cross-section of the hydrogen charge-exchange
      id=eiram_get_id('HYDHEL','H.1','3.1.8',err)
      if(err.ne.0) STOP "ERROR while looking for HYDHEL H.1 3.1.8"

C     calculate cross-section as a function of beam energy Eb,
C     use vector function
      call eiram_calc1(sigma,id,log(Eb),err)
      if(err.ne.0) STOP "ERROR while calculating cross-section"
      sigma=exp(sigma)

C     ... calculate cross-section, use fast scalar function,
C     print results
      write(*,*)
      write(*,*) 'HYDHEL H.1 3.1.8'
      write(*,*) ' sigma [cm-3], E [eV]'
      do k=1,L
```

```
        sigma(k)=eiram_calc1_fast(id,log(Eb(k)))
        write(*,'(1X,E10.2,1X,F7.1)') sigma(k),Eb(k)
      end do

C ##########################################################################
C  Calculation heavy particle collision rate
C  (depend on both energy a background temperature)
C ##########################################################################

C     take Id of reaction H.3 3.1.8 -
C     - collision rate of the hydrogen charge-exchange
      id=eiram_get_id('HYDHEL','H.3','3.1.8',err)
      if(err.ne.0) STOP "ERROR while looking for HYDHEL H.3 3.1.8"

C     calculate fit coefficients reduced over second variable
      call eiram_get_order(N,M,id,err)
      if(err.ne.0) STOP "ERROR detected in  eiram_get_order"
      allocate(B(N+1,L))
      call eiram_calc12(B,id,log(Ti),err)
      if(err.ne.0) STOP "ERROR while calculating fit coefficients"

C     calculate collision rates for E=Eb(2), print results
      lnE=log(Eb(2))
      write(*,*)
      write(*,'(1X,A,F5.1,A)') 'HYDHEL H.3 3.1.8, E=',Eb(2),' [eV]'
      write(*,*) ' rate [cm3/s], Ti [eV]'
      do k=1,L
        rate(k)=exp(eiram_fit(B(:,k),N,lnE))
        write(*,'(1X,E10.2,1X,F7.1)') rate(k),Ti(k)
      end do
      write(*,*)
      deallocate(B)

C ##########################################################################
C  Calculation of density dependent electron impact rate
C ##########################################################################

C     Append AMJUEL to already loaded HYDHEL
      call eiram_load('../data/','AMJUEL',err)
      if(err.ne.0) STOP "ERROR while loading AMJUEL"
```

```
C     take Id of reaction H.4 2.1.5 -
C     - ionization rate of hydrogen atoms
      id=eiram_get_id('AMJUEL','H.4','2.1.5',err)
      if(err.ne.0) STOP "ERROR while looking for AMJUEL H.4 2.1.5"

C     calculate collision rates for pairs (ne,Te=Ti)
C     Don't forget to divide density in cm-3 by 1e8!
      call eiram_calc2(rate,id,log(ne/1e8),log(Ti),err)
      if(err.ne.0) STOP "ERROR while calculating collision rates"

C     print results
      write(*,*)
      write(*,*) 'AMJUEL H.4 2.1.5'
      write(*,*) ' rate [cm3/s], ne [cm-3], Te [eV]'
      do k=1,L
       write(*,'(1X,2E10.2,1X,F7.1)') rate(k),ne(k),Ti(k)
      end do
      write(*,*)

C     deallocate memory
      call eiram_deallocate(err)

      STOP "EIRAM_EXAMPLE COMPLETED"

      end program eiram_example
```